TD 7 : Équations différentielles

Lancer Firefox et aller sur sagemath.ens.uvsq.fr. Se connecter sur Sage, et créer une nouvelle feuille sous le nom habituel. Commencer par rentrer la commande `automatic_names(True)`.

Vous devez mettre des commentaires pour indiquer tout ce qu'on fait.

Exercice 1. Il s'agit d'étudier l'équation différentielle $y' = -2ty$.

1. On va commencer par dessiner le champ des vecteurs tangents correspondant.
   Pour $t = t_0$, on a le point $P_0 = (t_0, y_0)$ où $y_0 = y(t_0)$. On sait qu'un vecteur tangent en P_0 est $(1, y'(t_0))$, on le normalise pour avoir un vecteur tangent unitaire $\vec{u}(t_0) = \left(\frac{1}{\sqrt{1+y'(t_0)^2}}, \frac{y'(t_0)}{\sqrt{1+y'(t_0)^2}}\right)$.
   Dans Sage, on va utiliser `y1` pour noter y'. On va appeler `DE` l'équation différentielle :

```
DE = y1==-2*t*y; DE
y1sols = solve(DE,y1); y1sols
y1sol = rhs(y1sols[0]); y1sol
vt = (1,y1sol);vt
p1 = plot_vector_field(vt,(t,-3,3),(y,-3,3)); p1
ut = (1/sqrt(1+y1sol^2),y1sol/sqrt(1+y1sol^2)); ut
p2 = plot_vector_field(vt,(t,-3,3),(y,-3,3)); p2
```

   Les solutions de l'équation différentielle sont donc les courbes qui "suivent" les chemins indiqués par les vecteurs du champ.

2. On va maintenant résoudre l'équation différentielle.
   Dans Sage, il faut déclarer t comme une variable et y comme une fonction de t. La commande à utiliser est `desolve`.

```
var('t')
y = function('y',t); y
y1 = diff(y,t); y1
DE = y1==-2*t*y; DE
DEres = desolve(DE,y,ivar=t,show_method=True); DEres
ysol = DEres[0]; ysol
```

   On voit que les solutions dépendent d'une constante c. On veut maintenant dessiner ces solutions en faisant varier c. Le problème est que cette variable n'est pas la variable c normale, mais une variable "interne" à Sage ; il faut donc quelques contorsions pour accéder à cette variable :

```
ysol.variables()
cc = ysol.variables()[0]; cc
var('ccc')
p2 = plot([ysol.subs(cc==ccc) for ccc in srange(-3, 3, 0.5)],
     (t, -3, 3),ymin=-3,ymax=3); p2
```

   On superpose les deux graphiques :

```
p1+p2
```

   et l'on voit bien comment les solutions "suivent" les vecteurs du champ.

3. On s'intéresse maintenant à une solution particulière, celle qui passe par le point $P_0 = (t_0, y_0) = (-2, 1/20)$.

```
(t0,y0) = (-2,1/20); P0 = [t0,y0]; P0
p0 = point(P0,color='red'); p0+p1
DEres0 = desolve(DE,y,ivar=t,ics=[t0,y0],show_method=True); DEres0
ysol0 = DEres0[0]; ysol0
p20 = plot(ysol0,(t,t0,3),ymin=-3,ymax=3); p0+p20
p1+p0+p20
```

Exercice 2. En suivant la démarche de l'exercice précédent, traiter les équations différentielles suivantes : $y' = y/t$, $y' = -y/t$ et $y' = y \tan t$, avec la condition initiale $y(0) = 1$.

Exercice 3. Il s'agit de résoudre le système $\begin{cases} x' = 2y + 2z \\ y' = -x + 2y + 2z \\ z' = -x + y + 3z \end{cases}$ avec la commande `desolve_system`.

```
var('t'); x=function('x',t); y=function('y',t); z=function('z',t)
x,y,z
x1=diff(x,t); y1=diff(y,t); z1=diff(z,t)
x1,y1,z1
de1=x1==2*y+2*z; de2=y1==-x+2*y+2*z; de3=z1==-x+y+3*z
de1,de2,de3
DESres=desolve_system([de1,de2,de3],[x,y,z],ivar=t);DESres
for s in DESres: s
```

Pour avoir la solution correspondante aux conditions initiales $x(t_0) = x_0$, $y(t_0) = y_0$, $z(t_0) = z_0$:

```
t0,x0,y0,z0 = 0,1,2,3; t0,x0,y0,z0
DESres0=desolve_system([de1,de2,de3],[x,y,z],ivar=t,ics=[t0,x0,y0,z0]);DESres0
for s in DESres0: s
```

On obtient une courbe dans l'espace, dont le dessin est donné par :

```
courbe=map(rhs,DESres0);courbe
parametric_plot3d(courbe,(t,-1,1))
```

Exercice 4. Résoudre le système $\begin{cases} (t^2+1)x' = tx + y + 2t^2 - 1 \\ (t^2+1)y' = x - ty + 3t \end{cases}$

Exercice 5. On considère l'équation différentielle (E) : $\begin{cases} y' = xy(y-1) \\ y(0) = a. \end{cases}$

1. Résoudre l'équation à l'aide de la commande `desolve`. Qu'est-ce que cela renvoie ? Est-on convaincu du résultat ? Qu'est-ce que Sage n'a pas su faire ?
2. Résoudre l'équation (E) pour les valeurs particulières de a : -1 ; 0 ; $0{,}5$; $0{,}8$; 1 ; 2. Obtient-on toujours la même solution ? Peut-on expliquer pourquoi ?
3. Résoudre à l'aide de Sage l'équation (E') : $\ln\left(\frac{y-1}{y}\right) = \frac{1}{2}x^2 + \ln(a-1) - \ln(a)$. Qu'est-ce que cela renvoie ? Vérifier que c'est bien la solution de (E).
4. Sur un même graphique, à l'aide de `plot`, tracer les solutions de (E') sur $[-1{,}1; 1{,}1]$ pour les valeurs de a suivantes : -1 ; 0 ; $0{,}5$; $0{,}8$; 1 ; 2. Les solutions se coupent-elles ? Pourquoi ? Que constate-t-on sur la monotonie et les régions du plans ? Est-ce en accord avec l'équation ?
5. Étudier l'évolution du graphique des solutions lorsque a est supérieur à 1 et se rapproche de 1.
6. Étudier l'évolution du graphique des solutions lorsque a est inférieur à 0 et se rapproche de 0.

Exercice 6. On s'intéresse à l'équation différentielle $(x-2)y' = 2y - x$. Tracer sur un même graphique les solutions de l'équation avec la condition $y(0) = a$ pour a prenant les valeurs entières entre -4 et 4. Que constate-t-on ?

Exercice 7. Dans cet exercice, nous allons résoudre des équations du second ordre.

1. Comment peut-on demander à Sage la dérivée seconde de y par rapport à x ?
2. Utiliser cette commande pour résoudre les équations différentielles suivantes :
   a. $y'' - 3y' + 2y = \mathrm{e}^x$;
   b. $y'' + 4y = \cos(2x) + \cos(4x)$;
   c. $y'' - y = 4x\mathrm{e}^x$;
   d. $y'' + 9y = t\cos x$;
   e. $y'' - y = \sin^2 x$;
   f. $y'' - y' + y = \sin(2x)$.
3. Comment peut-on faire pour spécifier à Sage que l'on veut la solution y telle que $y(0) = 1$ et $y'(0) = 2$?
4. Vérifier votre méthode sur les 6 équations précédentes.
5. Demander à Sage de résoudre les équations différentielles $y'' + xy' = 0$ et $y' + xy = 0$.
6. Quel doit être le lien entre les deux solutions ? En déduire ce qu'est la fonction `erf`.

Exercice 8. Soit a un réel positif et posons $(E_a)\quad y' = (xy)^a$. Dressez suivant les valeurs de a le portrait de phases des solutions. Sont-elles toujours définies sur $\mathbb{R}$. Se coupent-elles ? Discutez en fonction de a de la limite en $+\infty$ et de la monotonie des solutions.